

# NAG Library Routine Document

## X03ABF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

X03ABF calculates the value of a *complex\*16* scalar product using *basic precision* or *additional precision* and adds it to a *complex\*16* initial value.

### 2 Specification

```

SUBROUTINE X03ABF(A, ISIZEA, B, ISIZEB, N, ISTEPA, ISTEPB, CX, DX, SW,
1              IFAIL)
INTEGER          ISIZEA, ISIZEB, N, ISTEPA, ISTEPB, IFAIL
complex*16     A(ISIZEA), B(ISIZEB), CX, DX
LOGICAL          SW

```

### 3 Description

X03ABF calculates the scalar product of two *complex\*16* vectors and adds it to an initial value  $c$  to give a correctly rounded result  $d$ :

$$d = c + \sum_{i=1}^n a_i b_i.$$

If  $n < 1$ ,  $d = c$ .

The vector elements  $a_i$  and  $b_i$  are stored in selected elements of the one-dimensional array parameters A and B, which in the subroutine from which X03ABF is called may be identified with parts of possibly multi-dimensional arrays according to the standard Fortran rules. For example, the vectors may be parts of a row or column of a matrix. See Section 5 for details, and Section 9 for an example.

The products are accumulated in *basic precision* or *additional precision* depending on the parameter SW.

This routine has been designed primarily for use as an auxiliary routine by other routines in the NAG Library, especially those in the chapters on Linear Algebra.

### 4 References

None.

### 5 Parameters

1: A(ISIZEA) – *complex\*16* array *Input*

*On entry:* the elements of the first vector.

The  $i$ th vector element is stored in the array element  $A((i - 1) \times \text{ISTEPA} + 1)$ . In your subroutine from which X03ABF is called, A can be part of a multi-dimensional array and the actual argument must be the array element containing the first vector element.

2: ISIZEA – INTEGER *Input*

*On entry:* the dimension of the array A as declared in the (sub)program from which X03ABF is called.

The upper bound for ISIZEA is found by multiplying together the dimensions of A as declared in your subroutine from which X03ABF is called, subtracting the starting position and adding 1.

*Constraint:*  $ISIZEA \geq (N - 1) \times ISTEPA + 1$ .

3: B(ISIZEB) – *complex\*16* array *Input*

*On entry:* the elements of the second vector.

The  $i$ th vector element is stored in the array element  $B((i - 1) \times ISTEPB + 1)$ . In your subroutine from which X03ABF is called, B can be part of a multi-dimensional array and the actual argument must be the array element containing the first vector element.

4: ISIZEB – INTEGER *Input*

*On entry:* the dimension of the array B as declared in the (sub)program from which X03ABF is called.

The upper bound for ISIZEB is found by multiplying together the dimensions of B as declared in your subroutine from which X03ABF is called, subtracting the starting position and adding 1.

*Constraint:*  $ISIZEB \geq (N - 1) \times ISTEPB + 1$ .

5: N – INTEGER *Input*

*On entry:*  $n$ , the number of elements in the scalar product.

6: ISTEPA – INTEGER *Input*

*On entry:* the step length between elements of the first vector in array A.

*Constraint:*  $ISTEPA > 0$ .

7: ISTEPB – INTEGER *Input*

*On entry:* the step length between elements of the second vector in array B.

*Constraint:*  $ISTEPB > 0$ .

8: CX – *complex\*16* *Input*

*On entry:* the initial value  $c$ .

9: DX – *complex\*16* *Output*

*On exit:* the result  $d$ .

10: SW – LOGICAL *Input*

*On entry:* the precision to be used in the calculation.

If SW = .TRUE., *additional precision*.

If SW = .FALSE., *basic precision*.

11: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $ISTEPA \leq 0$ ,  
or  $ISTEPB \leq 0$ .

$IFAIL = 2$

On entry,  $ISIZEA < (N - 1) \times ISTEPA + 1$ ,  
or  $ISIZEB < (N - 1) \times ISTEPB + 1$ .

## 7 Accuracy

If the calculation is in *additional precision*, the result is correct to full implementation accuracy provided that exceptionally severe cancellation does not occur in the summation. If the calculation is in *basic precision*, such accuracy cannot be guaranteed.

## 8 Further Comments

The time taken by X03ABF is approximately proportional to  $n$  and also depends on whether *basic precision* or *additional precision* is used.

## 9 Example

This example calculates the scalar product of the second column of the matrix  $A$  and the vector  $B$ , and add it to an initial value of  $1 + i$ , where

$$A = \begin{pmatrix} -1 & -i & 1 \\ 2 + 3i & i & 2i \\ 0 & -1 - i & 1 - 2i \end{pmatrix}, \quad B = \begin{pmatrix} i \\ 1 - i \\ -i \end{pmatrix}.$$

### 9.1 Program Text

```
*      X03ABF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
INTEGER          N
PARAMETER       (N=3)
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
*      .. Local Scalars ..
COMPLEX *16     CX, DX
INTEGER          I, IFAIL, ISIZEA, ISIZEB, ISTEPA, ISTEPB, J
LOGICAL         SW
*      .. Local Arrays ..
COMPLEX *16     A(N,N), B(N)
*      .. External Subroutines ..
EXTERNAL        X03ABF
*      .. Executable Statements ..
WRITE (NOUT,*) 'X03ABF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) ((A(I,J),J=1,N),I=1,N), (B(I),I=1,N)
CX = (1.0D0,1.0D0)
ISIZEA = N
ISIZEB = N
ISTEPA = 1
ISTEPB = 1
SW = .TRUE.
```

```
      IFAIL = 0
*
      CALL X03ABF(A(1,2), ISIZEA, B, ISIZEB, N, ISTEPA, ISTEPB, CX, DX, SW, IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Result = ', DX
*
99999 FORMAT (1X,A,'(',F3.0,',',F3.0,')')
      END
```

## 9.2 Program Data

```
X03ABF Example Program Data
(-1.0, 0.0) ( 0.0, -1.0) (1.0,  0.0)
( 2.0, 3.0) ( 0.0,  1.0) (0.0,  2.0)
( 0.0, 0.0) (-1.0, -1.0) (1.0, -2.0)
( 0.0, 1.0) ( 1.0, -1.0) (0.0, -1.0)
```

## 9.3 Program Results

```
X03ABF Example Program Results
```

```
Result = ( 2., 3.)
```

---